



How Do You Use SAS® to Access Data and APIs From the Web?

Chris Hemedinger, SAS Online Communities





Chris Hemedinger

Senior Manager, SAS Online Communities

Chris leads the team of community managers for communities.sas.com, where SAS users gather and discuss SAS technologies, find answers to questions, share their experiences and provide feedback. He is also involved in SAS' strategy for other online communities with SAS customers. He gathers timely content for the user community, drawing more SAS experts into the conversations.

The Internet: Not just for web pages

And content is not just for web browsers

For end users



For automated processes



FILENAME URL

PROC HTTP

Any application that connects to a web site/service with HTTP is a web client

Today's topics

- Intro to FILENAME URL and PROC HTTP
- Use case: Download a file and read it with SAS
- Use case: Download a file with user/password
- Use case: Download and read a ZIP file in SAS
- Use case: Post data to a site via web form
- Use case: "Scrape" data from a web page
- Use case: Call REST API with JSON response



FILENAME URL

Simple. But limited.

```
filename data URL "https://sitewithdata.com/path-to-data";
```

Example:

```
filename data URL  
"https://www.federalreserve.gov/paymentsystems/files/coin_currcircvolume.txt";
```

- One line!
- Supports HTTP GET method only
- Supports basic options: user/password, proxy*, and a few others
- No control over where SAS stores the data that's fetched

PROC HTTP

More lines of code. More power.

```
proc http  
  method="method"  
  url="http://sitewithdata.com/path"  
  out=fileref;
```

Example:

```
filename data "/local/path/save-the-data-file";
```

```
proc http  
  method="GET"  
  url =  
    "https://www.federalreserve.gov/paymentsystems/files/coin_currcircvolume.txt"  
  out=data;  
run;
```

```
proc http
  method="method"
  url="http://sitewithdata.com/"
  out=fileref;
```

PROC HTTP

More lines. More power.

- Supports HTTP GET, POST, PUT, and any other method
- Provides (requires) control for location of output
- Many options for authentication, caching, cookie behavior – aspects of a full HTTP client
- Rich DEBUG support
- SAS language analogy to cURL

DEMO

Use case: Download a data file

Data from CMS.gov



```
filename nh temp;
```

```
proc http
```

```
url="https://data.cms.gov/api/views/s2uc-8wxp/rows.csv?accessType=DOWNLOAD"
```

```
method="GET"
```

```
out=nh;
```

```
run;
```

```
options validvarname=v7;
```

```
proc import file=nh
```

```
out=covid19nh
```

```
dbms=csv
```

```
replace;
```

```
run;
```


Download and import: Result

```
1 filename nh temp;
2 proc http
3   url="https://data.cms.gov/api/views/s2uc-8wxp/rows.csv?accessType=DOWNLOAD"
4   method="GET"
5   out=nh;
6 run;
7
8 options validvarname=v7;
9 proc import file=nh
10  out=covid19nh
11  dbms=csv
12  replace;
13 run;
14
15
```

Output Date (1)


> Where Query Builder Tasks

	Provider_Name	Provider_Address	Provider_City	Provider_State	Provider_Zip_Code	
1	LIFEQUEST NURSING CENTER	2450 JOHN FRIES HIGHWAY	QUAKERTOWN	PA	18951	Y
2	PENNYPACK NURSING AND REHABILITATION CENTER	8015 LAWNDALE AVENUE	PHILADELPHIA	PA	19111	Y
3	MEADOW VIEW NURSING CENTER	1404 HAY STREET	BERLIN	PA	15530	Y
4	ANN'S CHOICE	16000 ANN'S CHOICE WAY	WARMINSTER	PA	18974	Y
5	HAVENCREST HEALTHCARE AND REHABILITATION CENTER	1277 COUNTRY CLUB ROAD	MONONGAHELA	PA	15063	Y
6	LIFE CARE CENTER OF COLTEWAH	5911 SNOW HILL ROAD	COLTEWAH	TN	37363	Y
7	EDISTO POST ACUTE	575 STONEWALL JACKSON BOULEVARD	ORANGEBURG	SC	29115	Y
8	MAGNOLIA PLACE - SPARTANBURG	8020 WHITE AVENUE	SPARTANBURG	SC	29303	Y
9	GRAND STRAND REHAB AND NURSING CENTER LLC	4452 SOCASTEE BLVD	MYRTLE BEACH	SC	29588	Y
10	RIDGEVIEW TERRACE OF LIFE CARE	PO BOX 26 COFFEY LAKE	RUTLEDGE	TN	37861	Y
11	OAKBROOK HEALTH AND REHABILITATION CENTER	920 TRAVELERS BOULEVARD	SUMMERVILLE	SC	29405	Y
12	CALLAWAY NURSING HOME	1300 WEST LINDSEY	SULPHUR	OK	73095	Y
13	VALLEY WEST HEALTH CARE CENTER	2300 WARREN STREET	EUGENE	OR	97405	Y
14	PRESBYTERIAN COMMUNITIES OF SOUTH CAROLINA-SUMMERV	201 W 9TH NORTH STREET	SUMMERVILLE	SC	29403	Y
15	DYER NURSING AND REHABILITATION CENTER	1124 NORTH MAIN	DYER	TN	38330	Y
16	MUNCY PLACE	215 EAST WATER STREET	MUNCY	PA	17756	Y

DEMO

Use case: Download a file with user/password

Access a site with Basic authentication



```
proc http
```

```
url="www.secured-site.com"
```

```
out=resp
```

```
method="GET"
```

```
WEBUSERNAME="user"
```

```
WEBPASSWORD="pass"
```

```
AUTH_BASIC
```

```
AUTH_NEGOTIATE;
```

```
run;
```

Example of AUTH_BASIC

```
filename resp temp;
proc http
  url="http://httpbin.org/basic-auth/chris/pass123"
  method="GET"
  AUTH_BASIC
  out=resp
  webusername="chris"
  webpassword="pass123"
;
run;

data _null_;
  rc = jsonpp('resp','log');
run;
```

```
{
  "authenticated": true,
  "user": "chris"
}
```

Test methods using
<http://httpbin.org/>

DEMO



Tip: Using httpbin.org to test logic

```
filename resp "%sysfunc(getoption(WORK))/stream.json";
```

```
proc http
```

```
url="https://httpbin.org/stream/1"
```

```
method="GET"
```

```
out=resp;
```

```
run;
```

```
/* Supported with SAS 9.4 Maint 5 */
```

```
%put HTTP Status code = &SYS_PROCHTTP_STATUS_CODE. : &SYS_PROCHTTP_STATUS_PHRASE.;
```

```
data _null_;
```

```
rc = jsonpp('resp','log');
```

```
run;
```

```
HTTP Status code = 200 : OK
```

```
{  
  "url": "https://httpbin.org/stream/1",  
  "args": {  
  
  },  
  "headers": {  
    "Host": "httpbin.org",  
    "X-Amzn-Trace-Id": "Root=1-5f3bdd40-542722e0066832f012490708",  
    "User-Agent": "SAS/9",  
    "Accept": "*/*"  
  },  
  "origin": "199.199.188.99",  
  "id": 0  
}
```

DEMO

Use case: Download and read data in ZIP archive

Combine PROC HTTP and FILENAME ZIP

```
/* Download the ZIP file */  
filename dl "%sysfunc(getoption(WORK))/ri130701_13dn01.zip";  
  
proc http  
  url="http://www.freddiemac.com/mbs/data/stacr/ri130701_13dn01.zip"  
  method='GET'  
  out=dl  
  ct="application/zip";  
run;  
  
filename dl clear;  
  
/* Crack it open and read with FILENAME ZIP */  
filename inzip ZIP "%sysfunc(getoption(WORK))/ri130701_13dn01.zip";  
  
data orig;  
  infile inzip(ri130701_13dn01.txt);  
...
```

DEMO

Use case: Post data to the web via form

Customer name:

Telephone:

E-mail address:

Pizza Size

Small

Medium

Large

Pizza Toppings

Bacon

Extra Cheese

Onion

Mushroom

Preferred delivery time:

Delivery instructions:

```
filename resp temp;
proc http
  url="http://httpbin.org/post"
  method="POST"
  in="custname=Chris&size=large&topping=cheese"
  out=resp;
run;

data _null_;
  rc = jsonpp('resp','log');
run;
```

```
"form": {
  "custname": "Chris",
  "size": "large",
  "topping": "cheese"
},
```

DEMO

Use case: Scrape data from the web

But first, an editorial

Web scraping is lossy, fragile process.

The information on the web page does not include data types, lengths, or constraints metadata.

And one tweak to the presentation of the web page can break any automated scraping process.

Use case: Scrape data from the web

But if you must...

For this step	Use these features
Get the contents of the web page	PROC HTTP or FILENAME URL
Process/parse the web page contents	DATA step, with parsing functions such as FIND, SCAN , and regular expressions via PRXMATCH . Use SAS informats to convert text to native data types.
Repeat across subsequent pages	SAS macro language (%DO %UNTIL processing) or DATA step with CALL EXECUTE to generate multiple iterations of the fetch/parse steps.

<https://blogs.sas.com/content/sasdummys/scrape-web-page-data/>

Example of "web scraping"

National Notifiable Diseases Surveillance System (NNDSS)

COVID-19 Resources

Search Results for All Conditions

Search Conditions: Search All Conditions

Notifiable Condition Lists

Name	Acronym	Common Name	Etiology	Notifiable Years	Notifiable %
Acanthamoeba (excluding keratitis)					
Acanthamoeba keratitis					
Acquired immunodeficiency syndrome	AIDS			2000	2000
Acute Flaccid Paralysis (AFP)	AFP				
Arthritis		Encephalomyelitis		1920	1994
Asymptomatic pharyngitis/bacteriuria				2000	Current
Dementia		Bacterial meningitis		1944	Current
Enteric fever, non-typhoidal and typhoid				2005	Current

```
<table class="text-centered table" style="margin-bottom: 8px; width: 100%;">| Acanthamoeba (excluding keratitis) | | | | | |
| --- | --- | --- | --- | --- | --- |
|  |  |  |  |  |  |
| Acanthamoeba (excluding keratitis) | | | | | |
| Acanthamoeba keratitis | | | | | |
| Acquired immunodeficiency syndrome | | | | | |
| Acute Flaccid Paralysis (AFP) | | | | | |
| Arthritis | | | | | |
| Asymptomatic pharyngitis/bacteriuria | | | | | |
| Dementia | | | | | |
| Enteric fever, non-typhoidal and typhoid | | | | | |

```

DEMO

```
/* Get all of the nonblank lines */
filename CDC url "https://wwwn.cdc.gov/nndss/conditions/search/";
data rep;
infile CDC length=len lrecl=32767;
input line $varying32767. len;
line = strip(line);
if len>0;
run;
filename CDC clear;

/* Parse the lines and keep just condition names */
/* When a condition code is found, grab the line following (full name of condition) */
/* and the 8th line following (Notification To Date) */
/* Relies on this page's exact layout and line break scheme */
data parsed (keep=condition_code condition_full note_to);
length condition_code $ 40 condition_full $ 60;
set rep;
if find(line, "/nndss/conditions/") then do;
condition_code=scan(line,4,');
pickup=_n_+1;
pickup2=_n_+8;
set rep (rename=(line=condition_full)) point=pickup;
set rep (rename=(line=note_to)) point=pickup2;
output;
end;
run;
```



Use case: REST API with JSON response

```
/* Neat service from Open Notify project */  
filename resp temp;  
proc http  
  url="http://api.open-notify.org/astros.json"  
  method= "GET"  
  out=resp;  
run;  
  
data _null_ ;  
rc = jsonpp('resp','log');  
run;  
  
/* Assign a JSON library to the HTTP response */  
libname space JSON fileref=resp;
```

```
{  
  "number": 3,  
  "people": [  
    {  
      "craft": "ISS",  
      "name": "Chris Cassidy"  
    },  
    {  
      "craft": "ISS",  
      "name": "Anatoly Ivanishin"  
    },  
    {  
      "craft": "ISS",  
      "name": "Ivan Vagner"  
    }  
  ],  
  "message": "success"  
}
```

```
▼ SPACE  
  > ALLDATA  
  ▼ PEOPLE  
    ▲ craft  
    ▲ name  
    ⊕ ordinal_people  
    ⊕ ordinal_root  
  > ROOT
```

DEMO

Other functions to help

HTMLDECODE Function	Decodes a string that contains HTML numeric character references or HTML character entity references and returns the decoded string.
HTMLENCODE Function	Encodes characters using HTML character entity references and returns the encoded string.
URLDECODE Function	Returns a string that was decoded using the URL escape syntax.
URLENCODE Function	Returns a string that was encoded using the URL escape syntax. From: <code>urlencode(%str(ga:sessions,ga:pageviews,ga:users));</code> To: <code>ga%3Asessions,ga%3Apageviews,ga%3Ausers</code>
<code>%str(&)</code>	Prevent SAS from interpreting “&” in a URL as a macro variable. <code>in="custname=Joe%str(&)size=large%str(&)topping=cheese"</code> Avoids "WARNING: Apparent symbolic reference TOPPING not resolved."
JSONPP function	"Pretty print" a JSON response to a file or log

Learn more


PROC HTTP on SAS blogs

<https://blogs.sas.com/content/tag/proc-http/>

SAS Global Forum papers

[ABCs of PROC HTTP](#)

[REST Just Got Easy with SAS and PROC HTTP](#)



Q & A

sas.com



Explore Helpful Resources

[Ask the Expert](#)

View other user webinars that provide insights into using SAS products to make your job easier.

[FREE Training](#)

Learn from home – free for 30 days. Get software labs to practice and online support if needed.

[SAS Support Communities](#)

Ask questions, get answers and share insights with SAS users.

[SAS Analytics Explorers](#)

An exclusive platform to collaborate, learn and share your expertise. Gain access to a diverse network to advance your career. Special rewards and recognition exclusively for SAS users.

[SAS Users YouTube Channel](#)

A plethora of videos on hundreds of topics, just for SAS users.

[Newsletters](#)

Get the latest SAS news plus tips, tricks and more.

[Users Groups](#)

Meet local SAS users, network and exchange ideas – virtually.

[SAS Profile](#)

If you haven't already done so, create your SAS Profile to access free training, SAS Support Communities, technical support, software downloads, newsletters and more.

Did you know?

SAS is the only vendor named a Leader in the top 4 analyst assessments focused on the AI and Analytics market in 2020.



Learn more at sas.com/leader

